Creating a Context Index; how simple can it get?

Overview 
This presentation will look at what happens when we have a note or comment column in a table and we put an index on it so we can query it faster. It will explain how queries that use this column in the where clause will get full table scans even though we have a regular index on it. We will explain how a context index works. 

There will be a simple example that shows the steps to build a context index on a note field which uses a DIRECT_DATASTORE. It will cover what tables are created after the index is created. To better clarify what is happening we will provided information on how Oracle stores and uses indexes. We will also cover the different types of Datastores that are available with Oracles context indexes. We cover how Oracle Text can be created on a single column in a table; multiple columns in a table or tables; column containing a URL; or a column contain a file location.  

We will discuss the different preferences that are used when creating a context index.  We will cover how we know that the context index needs to be resynced and if any errors occurred during the indexing process. 

Oracle Text -- What is it?

Oracle Text is an Oracle tool that helps us put indexes on text columns or index documents or URL’s.  An Oracle Context index is not a single object, but a combination of multiple objects that are used to store and retrieve data.  Oracle Text is a tool that enables you to build text query applications and document classification applications. Oracle Text provides indexing, word and theme searching, and viewing capabilities for text.

We all want our data fast, and we have all experienced the scenario when our customers or developers want a note or a comment field in a table.   This note can be as small as 20 characters or as large as 4000 characters or maybe even a blob or a clob.   So we add a column called Note or Comment to our table.   Then we hear that they want to be able to query the column based on a word or phrase.   Which leads us to put an index on the column.   That sounds like a great idea, but what is it really doing and will it work?

Let go through a simple example.   Let say we have a table called;  PARTS that tracks parts for a supply house.   They want to have a note field that can be used to put any kind of note.   Let say that we have a PART called, “trampoline Spring”.   Now someone adds a free text note about the TRAMPOLINE SPRING.    Let’s say that they add a comment about the size, description and material of the SPRING.

Now months later we need to query the table and retrieve the record.  We don’t know where the word SPRING resides in the column so we will have to wild card in the query.
Our query would look something like this:

Select * from PARTS 

Where UPPER(note) like ‘%SPRING%’;

Let’s even simplify this and say we can remove the UPPER function because we store all of the text in upper case format.

Select * from PARTS 

Where note like ‘%SPRING%’;

Let’s also say that this table has a hundred thousands records.  What is going to happen when we perform the query?

Let’s look at the explain plan.

-
SELECT STATEMENT

-  TABLE ACCESS  (FULL)  -- PARTS (TABLE)

Why the full table scan?    Because we have to do a character by character search through every record.

This is exactly what Oracle text and Context indexes was designed to do.

If we had a context index on this column we could write the following SQL statement and get an index on the word “SPRING”.

Here is the sql statement:

Select * from PARTS 

Where CONTAINS(note, ‘spring’) > 0;     --  simplest without assigning a label

Select score(1) , a.*  from PARTS a

Where CONTAINS(note, ‘spring’,1) > 0;

Now what does the explain plan look like:

-
SELECT STATEMENT

-  TABLE ACCESS  (BY INDEX ROWID)  -- PARTS (TABLE)

-     DOMAIN INDEX -  PARTS_NOTE_CTX_IDX(INDEX (DOMAIN)) 

Now that is a better explain plan and the query will return in under a second as apposed to minutes. 
On what kind of column can you put a context Index?

The following columns types are supported: CHAR, VARCHAR, VARCHAR2, BLOB, CLOB, BFILE, or XMLType. 

Oracle Text uses different types of DATASTOREs depending on how your data is stored.   Here are the options:

Datastore Type

Use When
DIRECT_DATASTORE
Data is stored internally in the text column. Each row is indexed as a single document. (No Attributes)

MULTI_COLUMN_DATASTORE
Data is stored in a text table in more than one column. Columns are concatenated to create a virtual document, one per row. 

DETAIL_DATASTORE
Data is stored internally in the text column. Document consists of one or more rows stored in a text column in a detail table, with header 
information stored in a master table. (Master - Detail tables)

FILE_DATASTORE
Data is stored externally in operating system files. Filenames are stored in the text column, one per row.

NESTED_DATASTORE
Data is stored in a nested table. 
URL_DATASTORE 
Data is stored externally in files located on an intranet or the Internet. Uniform Resource Locators (URLs) are stored in the text column. 

USER_DATASTORE
Documents are synthesized at index time by a user-defined stored procedure. 

How do we create a context index?

Here is our simple table:

create table PARTS

(


PARTNUMBER number primary key, 


DESCRIPTION VARCHAR2(100),

             SUPPLIER_ID  NUMBER

); 

.

.

.

insert into PARTS values(111555,'12 Volt Battery',1010;

insert into PARTS values(111556,'3/4 screen matt Black', 1001);

insert into PARTS values(111556,'Trampoline Spring, 10 inch, galvanized 12 guage’, 2032);

.

.

.

commit;

Example:  DIRECT_DATASTORE

create index PARTS_NOTE_CTX_IDX on PARTS(NOTE) 

  indextype is ctxsys.context 

  parameters ('DATASTORE CTXSYS.DEFAULT_DATASTORE');

That is the simplest form of the “CREATE CONTEXT INDEX”.

You can instantly query right after this command.  

Now here is your basic query.

Select * from PARTS 

Where CONTAINS(note, ‘spring’) > 0;     

The “where” clause with a context query can be used like any other where clause with and’s and or’s.

We are basically taking all of the defaults.   Let step a little deeper into the creation of this index and how to use it.

1.
What options are available?

2.
What is created when we create the index?

3.
How to query the table using the context index?

What kind of datastores are available:

1.          Direct_Datastore

2.
Multi-column_Datastore

3.
Detail_Datastore

4.
File_Datastore

5.
Nested_Datastore

6.
URL_Datastore

7.
User_Datastore

The first part of setting up a context index is to decide which type of data store that you will be using.   For our example we will be using a Direct Datastore.

What tables are created when a context index is created?

Once you create a context index, you will notice that you have several new tables.   The best way to see these tables is with the following query.  

select table_name from user_tables where tablename like 'DR$%';

TABLE_NAME

DR$PARTS_NOTE_CTX_IDX$I

DR$PARTS_NOTE_CTX_IDX$N

DR$PARTS_NOTE_CTX_IDX$K

DR$PARTS_NOTE_CTX_IDX$R

These are the tables that will be created.  They are not indexes!  An Oracle Text index in not a single object, but a combination of multiple objects.   These objects start with DR$, then the index name, then the suffix; for example K$.  The tables are as follows with an explanation of what they do and contain.

dr$index_name$i
This table is the index table or "Token" table.   It contains all of the token words that have been indexed, with relative information about the document and where they occurred within the document.  Each document is assigned an internal document_id.

dr$index_name$k
This table is the keymap table.  It is an indexed-organized table that contains the document_id and maps it to a rowid.

dr$index_name$n
This table is a negative row table.   It contains a list of deleted document_ids.  (These are values that are used for cleanup during the optimization of the index.

dr$index_name$p
An index-organized table that is created when SUBSTRING_INDEX is enabled in the BASIC_WORDLIST.

dr$index_name$r
This table is the ROWID table and is designed for the opposite lookup from the $K table.   Unlike the $K table, however, a single row can contain multiple ROWIDs.

dr$index_name$x
This is the index for the $i table.


Only the $I, $N, $K and $R objects will be created every time and the $P and $X objects will be created based on which type of index is created and the preferences.

To help manage the context index, we can put the different objects in different tablespaces and different locations.   You can put everything in the same tablespace, but you will get better performance if you split them apart on different drives.  

Index Preferences and Attributes
You can always just take the defaults which are the easiest ways to create a context index, however it is recommended that you create your own preferences and override the defaults.  These are the preferences that you can define:

· Datastore:   This preference indicates what type of datastore to use.

· Filter:  This allows full-text documents.  These documents can be word processor documents, PDF, XML, HTML or plain text.

· Lexer:  Specifies the language of the text to be indexed.

· Wordlist:   Used to enable stemming and fuzzy matching query options, substring and prefix indexing, and to set the maximum number of terms in a wild card expression.

· Storage:  Used to specify the tablespaces and other storage parameters to be used when creating the tables, and indexes that are associated with the text index.

· Seciton Groups:   Used in order to be able to issue WITHIN queries on sections within a document or table columns.

· Stoplists:  Used to define words that are not to be indexed.

Attributes give more definition to Preferences.   We will see how later.

Since Oracle Text is an Oracle tool you will need permissions to run it.

First things first.    In order to setup a datastore, we need to grant the following to the user that will be creating the index.  

/****   This needs to be run by ctxsys   ***** */

grant ctxapp to dad_owner;      --   dad_owner is the owner of our application.

/

grant execute on ctx_ddl to dad_owner;

/

/****   done   ***** */

--  From now on you should be able to do every command through the dad_owner account.

Index Preferences
We will be using DIRECT_DATASTORE which is the default for datastore.  Let's get busy and create our first preference.   I am showing the way to delete a preference even though we may not have one created yet.   It is a good starting point, so just ignore the error if it says that there isn't one.

Exec Ctx_Ddl.Drop_Preference( 'parts_datastore' );

Exec Ctx_Ddl.Drop_Preference( 'parts_filter' );

Exec Ctx_Ddl.Drop_Preference( 'parts_lexer' );

Exec Ctx_Ddl.Drop_Preference( 'parts_wordlist' );

Exec Ctx_Ddl.Drop_Preference( 'parts_storage' );

Exec Ctx_Ddl.Drop_Section_Group( 'parts_section_group' );

Exec Ctx_Ddl.Drop_Stoplist( 'parts_stoplist' );

This will let you see what preferences we have available.

select pre_name name, 'preference' type from ctx_user_preferences

union all

select sgp_name name, 'section_group' type from ctx_user_section_groups

union all

select spl_name name, 'stop_list' type from ctx_user_stoplists;

Now let’s starting creating our preferences.

Begin


ctx_ddl.create_preference('parts_datastore','DIRECT_DATASTORE);

end;

/

Lexer
We will be creating a basic lexer that will not have any themes and list other white space character besides a <space> and a <tab>.  We do this by creating our PARTS_LEXER.   We will also create some attributes for our PARTS_LEXER.

Begin


ctx_ddl.create_preference('parts_lexer', 'BASIC_LEXER') ;


ctx_ddl.set_attribute('parts_lexer', 'INDEX_THEMES', 'NO') ;


ctx_ddl.set_attribute('parts_lexer', 'WHITESPACE',



'~!#$%^&()_=+[{]}\|;:",<>/?*.');


ctx_ddl.set_attribute('parts_lexer', 'SKIPJOINS', '`-''') ;

End ;

/

Wordlists

Skip WORDLISTS  This section will be commented out and will not be executed.

/*

Begin


ctx_ddl.create_preference('parts_wordlist', 'BASIC_WORDLIST') ;


ctx_ddl.set_attribute('parts_wordlist', 'STEMMER', ‘ENGLISH’) ;

ctx_ddl.set_attribute('parts_wordlist', 'FUZZY_MATCH', ‘ENGLISH’) ;

ctx_ddl.set_attribute('parts_wordlist', 'FUZZY_SCORE', ‘’0’) ;

ctx_ddl.set_attribute('parts_wordlist', 'FUZZY_SCORE', ‘’5000’) ;

ctx_ddl.set_attribute('parts_wordlist', 'SUBSTRING_INDEX', ‘YES’) ;

ctx_ddl.set_attribute('parts_wordlist', 'WILDCARD_MAXTERMS', 15000) ;

End ;

/

*/

Storage
This section will allow us to put our indexes in other tablespaces.

/*The STORAGE preference is used to specify the tablespace and other storage parameters to be used when creating the tables and indexes that are associated with the Text index. If this preference is not defined these objects are created in the index owner's default tablespace.

There is only one STORAGE type called BASIC_STORAGE.    The BASIC_STORAGE has the following attributes:

•
i_table_clause – The storage parameters for the dr$indexname$i table creation.

•
k_table_clause – The storage parameters for the dr$indexname$k table creation.

•
r_table_clause – The storage parameters for the dr$indexname$r table creation.

•
n_table_clause – The storage parameters for the dr$indexname$n table creation.

•
i_index_clause – The storage parameters for the dr$indexname$x index creation.

•
p_table_clause – The storage parameters for the dr$indexname$p table creation.

Metalink Note 179380.1, Bad Indexing and query performance on Intermedia Text indexes, recommends the following:

•
The r_table_clause must contain ' lob (data) store as (cache)'

•
The i_index_clause should contain ' compress 2 '

•       
Also recommend putting i_table and the x$index in different tablespaces from each other and the data.

*/

Begin


ctx_ddl.create_preference('parts_storage', 'BASIC_STORAGE') ;


ctx_ddl.set_attribute('parts_storage', 'I_TABLE_CLAUSE', 



'TABLESPACE sii_lob_indx01') ;


ctx_ddl.set_attribute('parts_storage', 'K_TABLE_CLAUSE', 



'TABLESPACE sii_lob1') ;


ctx_ddl.set_attribute('parts_storage', 'R_TABLE_CLAUSE', 



'TABLESPACE sii_lob_indx01



LOB (data) STORE AS (CACHE)') ;


ctx_ddl.set_attribute('parts_storage', 'N_TABLE_CLAUSE', 



'TABLESPACE sii_lob_indx01') ;


ctx_ddl.set_attribute('parts_storage', 'I_INDEX_CLAUSE', 



'TABLESPACE sii_lob1 COMPRESS 2') ;

End ;

/

Section Group

Skip SECTION GROUP    This section will be commented out and will not be executed.

/*

Begin

ctx_ddl.create_section_group('PARTs_section_group','BASIC_SECTION_GROUP') ;

ctx_ddl.add_field_section('PARTs_section_group', 'notes', 'notes', TRUE) ;

ctx_ddl.add_field_section('PARTs_section_group', 'part_note', 'pnsg', TRUE) ;

End ;

*/

Stop List 
Stop list section;   stoplist called PARTS_STOPLIST

Begin

ctx_ddl.create_stoplist('parts_stoplist', 'BASIC_STOPLIST') ;

cts_ddl.create_stopword('parts_stoplist', 'a');

cts_ddl.create_stopword('parts_stoplist', 'as');

cts_ddl.create_stopword('parts_stoplist', 'at');

cts_ddl.create_stopword('parts_stoplist', 'about');

……

……

-- This will not allow numbers to be indexed the parts_stoplist

ctx_ddl.add_stopclass('parts_stoplist', 'NUMBERS');       -- You have to decide if numbers are important to search.

End ;

/

Creating the Index
-- This will create the context for the table PARTS

CREATE INDEX PARTS_NOTE_CTX_IDX
ON PARTS(NOTES)

INDEXTYPE IS CTXSYS.CONTEXT

PARAMETERS('

DATASTORE                     PARTS_DATASTORE

LEXER                               PARTS_LEXER

-- SECTION GROUP         PARTS_SECTION_GROUP        no section group will be used.

STOPLIST                         PARTS_STOPLIST

STORAGE                         PARTS_STORAGE

-- WORDLIST                    PARTS_WORDLIST                       no wordlist will be used.

POPULATE');                                                                               --  nopopulate also available 
-- To view errors

> SELECT * FROM CTX_USER_INDEX_ERRORS;   
-- can use  where err_timestamp > sysdate;             -- for todays errors.
which uses the table CTXSYS.DR$INDEX_ERROR.    Which is a view and uses the table CTXSYS.DR$INDEX_ERROR.
Once you get an error and you fix it, you have to manually clean up this table yourself.

DELETE FROM ctx_user_index_errors;       -- you can use a where statement to remove specific errors.

Once you add records to the table that is indexed, the context index will not reflect the new record until you re-sync the context index.   

To see if you have any records that need to be re-synced, perform this query:

select * from ctx_user_pending;

Here is the syntax to resync the context index. 
begin

ctx_ddl.sync_index('PARTS_NOTE_CTX_IDX', '2M');

end; 

/

Always check to see that all records were indexed.

Query Context Index

When querying a context index we need to use the contains operator in the WHERE clause as part of the select statement.  Syntax:
CONTAINS(

         [schema.]column,

         text_query    VARCHAR2

         [,label       NUMBER])

RETURN NUMBER;

Example:

SELECT SCORE(1), partnumber, note from parts 

    WHERE CONTAINS(note, 'spring', 1) > 0  

order by score(1) desc;

Operators for Context Search

Examples for every CONTAINS operator. The following topics are covered:
ACCUMulate ( , )              ‘dog, cat, mouse’
AND ( & )                          ‘dog & cat  & mouse‘
MINUS ( - )                       ‘dog – poodles’
NEAR ( ; )                         ‘dog ; hunting’
NOT (~)                            ‘springs ~ plastic’
OR ( | )                              ‘tacks | staples’ 
soundex ( ! )                       ‘!smythe’               -- sounds like SMITH
stem ( $ )                            ‘$sing’                   -- expands to sing sang sung 
Exact string                        ‘cat and the fiddle’

Examples of Context Search
Example of AND: 
SELECT SCORE(1), partnumber, note from parts 

    WHERE CONTAINS(note, 'spring & copper', 1) > 0  

Example of AND, NEAR: 

SELECT SCORE(1), partnumber, note from parts 

    WHERE CONTAINS(note, 'spring & copper ; nickle ', 1) > 0  

Example of exact string: 
SELECT SCORE(1), partnumber, note from parts 

    WHERE CONTAINS(note, 'for fun it is a wonderful toy', 1) > 0  

Other Context commands that I found useful

You can also create a PARTITIONED CONTEXT INDEX.

Example of partitioned index:

CREATE INDEX DOC_REV_CTX_IDX ON DOCUMENT_REVISION(FILE_HTTP_LOCATION) INDEXTYPE IS ctxsys.context

   PARAMETERS('datastore PDRS_DOC_PREF 

               LEXER PDRS_LEXER 

               STOPLIST PDRS_STOPLIST 

               filter pdrs_custom_filter

               format column fmt

               nopopulate') 

   LOCAL (PARTITION DOC_CTX_IDX1, PARTITION DOC_CTX_IDX2, PARTITION DOC_CTX_IDX3, PARTITION DOC_CTX_IDX4,

          PARTITION DOC_CTX_IDX5, PARTITION DOC_CTX_IDX6, PARTITION DOC_CTX_IDX7, PARTITION DOC_CTX_IDX8,

          PARTITION DOC_CTX_IDX9, PARTITION DOC_CTX_IDX10, PARTITION DOC_CTX_IDX11, PARTITION DOC_CTX_IDX12)


PARALLEL 4;

-- This fixed partition 5 that was not completely built.   - resume is a new parameter for ctx indexes

alter index DOC_REV_CTX_IDX rebuild  PARTITION DOC_CTX_IDX5  parameters  ('resume') parallel 4;

-- you can also rebuild individual partitions

alter index DOC_REV_CTX_IDX rebuild  PARTITION DOC_CTX_IDX5;

Note 179380.1  Deals with your storage parameters.
Note 150453.1   Strategy for creating Oracle Text Index

If you are having issues with your Context index, you can turn on logging which will help you trouble shoot your issues.   Here are some of the commands that you would use to turn on logging:
Turn on index logging and check the output.
A log file can be created so that you can see the progress of Text index creation. The CTX_OUTPUT.START_LOG('<filename>') procedure begins logging to the file specified in '<filename>' and CTX_OUTPUT.END_LOG stops logging. By default, CTX_OUTPUT.START_LOG creates the specified log file in the $ORACLE_HOME/ctx/log directory. Alternatively, the default location for the Text index log files can be changed with the CTX_ADM procedure when connected as CTXSYS, as follows: 

connect CTXSYS/CTXSYS 

execute CTXSYS.CTX_ADM.SET_PARAMETER ('LOG_DIRECTORY','/tmp'); 

The user creating the index would start logging with the following command: 

execute  CTXSYS.CTX_OUTPUT.START_LOG('textidx.log'); 

When something goes wrong during index creation -- a core dump or a hang, for instance, you can usually work around it by removing or ignoring the faulty documents. In previous versions it was very difficult to find out which documents were faulty. Starting from 9i v9.0.1 CTX_OUTPUT has a new rowid printing event. When this is set the rowid of each row will be printed to the logfile. 

exec ctx_output.add_event(CTX_OUTPUT.EVENT_INDEX_PRINT_ROWID); 

In that case log would show: 

16:29:00 05/14/01 populate index: CTXSYS.LOGTESTX 

16:29:00 05/14/01 Begin document indexing 

16:29:00 05/14/01 INDEXING ROWID AAABklAACAAACynAAA 

16:29:00 05/14/01 INDEXING ROWID AAABklAACAAACynAAB 

execute  CTXSYS.CTX_OUTPUT.STOP_LOG; 

I have also used Oracle Context Index in several applications and multiple schemas.   To help in streamlining our processes and reuse our code, we wrote some shared procedures.  These procedures are called when we need to see if an record that was added to a table; which needs to causes a resync to the context index.  I will include the procedures that we wrote.   They should only be used as a guide.

CREATE OR REPLACE PROCEDURE DAD_OWNER.RESYNC_CTX_IDX(IDX_NAME in VARCHAR2, ERROR_MESSAGE out VARCHAR2)

AS

-- Author:  Michael Nelson

-- Date:  May 20, 2005

-- Purpose: This procedure will resync your context index immediately after inserting records.

-- It will check to see if any records are pending or waiting to be indexed.  Even if only

-- one record exits.  It will tell you if you have any error with your context index.

-- It requires two parameters one input variable that is the name of the index to be

-- resynced and the other will be the an output message if there are any error with the

-- index.   

--   

--  IMPORTANT:   Each schema using context search will have a copy of this procedure.

--

--  This procedure could easly be modified if we partitioned our context index and only resync

--  those partitions that need resyncing.

V_COUNT       NUMBER := 0;

V_ERROR_COUNT NUMBER := 0;

BEGIN

   SELECT COUNT(*) INTO V_COUNT

   FROM CTX_USER_PENDING

   WHERE PND_INDEX_NAME = UPPER(IDX_NAME)

   AND ROWNUM = 1;

   IF V_COUNT > 0 THEN

     CTXSYS.CTX_DDL.SYNC_INDEX(IDX_NAME,'2M');

   END IF;

   SELECT COUNT(*) INTO V_ERROR_COUNT

   FROM ctx_user_index_errors

   WHERE ERR_INDEX_NAME = UPPER(IDX_NAME)

   AND ROWNUM = 1;

   IF V_ERROR_COUNT > 0 THEN

--   message to be emailed to Administrator or Developer

     ERROR_MESSAGE := 'ERRORS on index ' || IDX_NAME || ' count is: ' || V_ERROR_COUNT;

   ELSE

     ERROR_MESSAGE := 'NONE';

   END IF;

END;

/

This next package is a common package that only needs to be written for one schema, but execute is granted to all other applications that use context indexes.

CREATE OR REPLACE PACKAGE SII_OWNER.Ctx_Idx_Error_Structure AS

/******************************************************************************

   NAME:       CTX_IDX_ERROR_STRUCTURE

   PURPOSE:    Provide a pipelined structure for passing of context indexing

               error data from *_OWNER to *_APP_USER.

   REVISIONS:

   Ver        Date           Authors                  Description

   ---------  --------------  -------------------------  ------------------------------------

   1.0        8/19/2005   Michael Nelson       1. Created this package.

******************************************************************************/

  TYPE CTX_IDX_ERROR_STRUCT IS RECORD

  (

    ERR_ROWID



VARCHAR2(18),

    ERR_INDEX_NAME 


VARCHAR2(30),

    ERR_TEXTKEY



VARCHAR2(18),

    ERR_TEXT 



VARCHAR2(4000),

    ERR_TIMESTAMP 


DATE

  );

  TYPE CTX_IDX_ERROR_TABLE IS TABLE OF CTX_IDX_ERROR_STRUCT;

END Ctx_Idx_Error_Structure;

/

GRANT EXECUTE ON SII_OWNER.CTX_IDX_ERROR_STRUCTURE TO "PUBLIC"

/

CREATE OR REPLACE PACKAGE PDRS_OWNER.Idx_Err_Interface
AS

--  This is used to pass data from %_owner to another user.
  FUNCTION get_index_errors(
    v_index_name          IN VARCHAR2 DEFAULT 'DOC_REV_CTX_IDX',
    v_err_text_srch           IN VARCHAR2 DEFAULT '%',
    v_rowid_srch        IN VARCHAR2 DEFAULT '%'
  )
  RETURN SII_OWNER.Ctx_Idx_Error_Structure.CTX_IDX_ERROR_TABLE PIPELINED;
END Idx_Err_Interface;

/

CREATE OR REPLACE PROCEDURE DAD_OWNER.syncTextIndex(index_name in varchar2, storage_size in varchar2)
authid current_user
AS
v_index_name varchar2(64) := index_name;
v_storage_size varchar2(64) := storage_size;
begin
   ctx_ddl.sync_index(v_index_name, v_storage_size);
end;

/

CREATE OR REPLACE FUNCTION DAD_OWNER.CTX_ERR_EXISTS(IDX_NAME IN VARCHAR2, ROW_XX IN VARCHAR2)
RETURN NUMBER IS

retVal NUMBER;
/******************************************************************************
   NAME:       CTX_ERR_EXISTS
   PURPOSE:    Returns 1 if at least one error exists in CTX_USER_INDEX_ERRORS
               related to a specific row for a given index name.  Returns 0
               if no errors exist.
               
   PARAMETERS:
               IDX_NAME - Name of the index table (for example, 'DOC_REV_CTX_IDX')
               ROW_XX   - ROWID of the document (GET_ROWID procedure should return
                          a usable value for this)
   REVISIONS:
   Ver        Date        Author                  Description
   ---------  ----------  ---------------  ------------------------------------
   1.0        9/15/2005   Michael Nelson  1. Created this function.

******************************************************************************/
BEGIN
   retVal := 0;

   SELECT COUNT(*) INTO retVal
   FROM ctx_user_index_errors
   WHERE ERR_INDEX_NAME = IDX_NAME
   AND ERR_TEXTKEY LIKE ROW_XX
   AND ROWNUM = 1;

   RETURN retVal;

   EXCEPTION
     WHEN NO_DATA_FOUND THEN
       NULL;
     WHEN OTHERS THEN
       -- Consider logging the error and then re-raise
       RAISE;
END CTX_ERR_EXISTS;
/

CREATE OR REPLACE FUNCTION DAD_OWNER.EXTRACT_ERR_SUBCODE(ERROR_TEXT in VARCHAR2) RETURN VARCHAR2 IS
retVal VARCHAR2(5);
/******************************************************************************
   NAME:       EXTRACT_ERR_SUBCODE 
   PURPOSE:    Grabs an error subcode from an oracle indexing error message.
   BACKGROUND: Some indexing errors will include an error subcode that we can use
         to more accurately identify the type of error we're encountering.
      For example: 
      "DRG-11207: user filter command exited with status 137" 
      The error subcode for the above would be 137.

      In reality, this function will return the last "word" in any parameter
      passed to it--that is, the piece of the text which follows the last  
      space.  Any trailing carriage-return characters will be trimmed off,
      and the "word" will be truncated to 5 characters. 

******************************************************************************/
BEGIN
   -- extract the bit of the string AFTER the last space,
   -- and trim off any trailing CR (carriage-return) characters 
   retVal := SUBSTR(RTRIM(SUBSTR(ERROR_TEXT, INSTR(ERROR_TEXT,' ', -1, 1)+1),CHR(10)),1,5);
  
   RETURN retVal;


   EXCEPTION
     WHEN NO_DATA_FOUND THEN
       NULL;
     WHEN OTHERS THEN
       -- Consider logging the error and then re-raise 
       RAISE;
END EXTRACT_ERR_SUBCODE;
/


CREATE OR REPLACE FUNCTION DAD_OWNER.Ctx_Output_Detected(IDX_NAME IN VARCHAR2, ROW_XX IN VARCHAR2)
RETURN NUMBER IS

retVal NUMBER;
V_TOKEN_COUNT NUMBER;

/******************************************************************************
   NAME:       CTX_OUTPUT_DETECTED
   PURPOSE:    Returns 1 if output was generated on the last attempt to index
   


   a specific row of an indexed table; otherwise returns 0.

   PARAMETERS:
               IDX_NAME - Name of the index table (for example, 'DOC_REV_CTX_IDX')



   ROW_XX   - ROWID of the document (GET_ROWID procedure should return



              a usable value for this)

   REVISIONS:
   Ver        Date        Authors                          Description
   ---------  ----------  -------------------------------  ------------------------------------
   1.0        9/15/2005   Bryan Bale / Michael Nelson      1. Created this function.

******************************************************************************/
BEGIN
   retVal := 0;

   EXECUTE IMMEDIATE

        'SELECT COUNT(COALESCE(i.TOKEN_FIRST, i.TOKEN_LAST))
   FROM


    DR$' || IDX_NAME || '$K k,


    DR$' || IDX_NAME || '$I i
   WHERE


    k.TEXTKEY = :THISROWID
   AND


    ((k.DOCID = i.TOKEN_FIRST) OR (k.DOCID = i.TOKEN_LAST))'
   INTO
   
     V_TOKEN_COUNT
   USING ROW_XX;

   IF V_TOKEN_COUNT > 0 THEN
      retVal := 1;
   END IF;

   RETURN retVal;

   EXCEPTION
     WHEN NO_DATA_FOUND THEN
       NULL;
     WHEN OTHERS THEN
       -- Consider logging the error and then re-raise
       RAISE;
END Ctx_Output_Detected;
/


SUMMARY

Next to its employees, information is a company’s biggest asset. How a company manages its data can be a strategic advantage or a disadvantage to its competitors.  Oracle Text enables text columns and documents to be search more efficiently.  By using a Direct Datastore, you can Oracle Context Indexes which speed up queries.  By just implementing a different datastore, a developer can use easily modify this presentation and examples to create a context index on a single column, multi-column, URL, or a file.  This will easily allow the user to index full-text documents.  One of the key advantages of Oracle text is that it allows you to implement full-text search functionality against data in your database without the addition of a third party tool.

